// File: CurveProxy.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

import "Address.sol";
import "SafeERC20.sol";
import "IGaugeController.sol";
import "ILiquidityGauge.sol";
import "PrismaOwnable.sol";

interface IVotingEscrow {
    function create_lock(uint256 amount, uint256 unlock_time) external;

    function increase_amount(uint256 amount) external;

    function increase_unlock_time(uint256 unlock_time) external;
}

interface IMinter {
    function mint(address gauge) external;
}

interface IFeeDistributor {
    function claim() external returns (uint256);

    function token() external view returns (address);
}

interface IAragon {
    function vote(uint256 _voteData, bool _supports, bool _executesIfDecided) external;
}

/**
    @title Prisma Curve Proxy
    @notice Locks CRV in Curve's `VotingEscrow` and interacts with various Curve
            contracts that require / provide benefit from the locked CRV position.
    @dev This contract cannot operate without approval in Curve's VotingEscrow
         smart wallet whitelist. See the Curve documentation for more info:
         https://docs.curve.fi/curve_dao/VotingEscrow/#smart-wallet-whitelist
 */
contract CurveProxy is PrismaOwnable {
    using Address for address;
    using SafeERC20 for IERC20;

    event CrvFeePctSet(uint256 feePct);

    IERC20 public immutable CRV;
    IGaugeController public immutable gaugeController;
    IMinter public immutable minter;
    IVotingEscrow public immutable votingEscrow;
    IFeeDistributor public immutable feeDistributor;
    IERC20 public immutable feeToken;

    uint256 constant WEEK = 604800;
    uint256 constant MAX_LOCK_DURATION = 4 * 365 * 86400; // 4 years

    uint64 public crvFeePct; // fee as a pct out of 10000
    uint64 public unlockTime;

    // the vote manager is approved to call voting-related functions
    // these functions are also callable directly by the owner
    address public voteManager;

    // the deposit manager is approved to call all gauge-related functionality
    // and can permit other contracts to access the same functions on a per-gauge basis
    address public depositManager;

    // permission for contracts which can call gauge-related functionality for a single gauge
    mapping(address caller => address gauge) public perGaugeApproval;

    // permission for callers which can execute arbitrary calls via this contract's `execute` function
    mapping(address caller => mapping(address target => mapping(bytes4 selector => bool))) executePermissions;

    struct GaugeWeightVote {
        address gauge;
        uint256 weight;
    }

    struct TokenBalance {
        IERC20 token;
        uint256 amount;
    }

    constructor(
        address _prismaCore,
        IERC20 _CRV,
        IGaugeController _gaugeController,
        IMinter _minter,
        IVotingEscrow _votingEscrow,
        IFeeDistributor _feeDistributor
    ) PrismaOwnable(_prismaCore) {
        CRV = _CRV;
        gaugeController = _gaugeController;
        minter = _minter;
        votingEscrow = _votingEscrow;
        feeDistributor = _feeDistributor;
        feeToken = IERC20(_feeDistributor.token());

        CRV.approve(address(votingEscrow), type(uint256).max);
    }

    modifier ownerOrVoteManager() {
        require(msg.sender == voteManager || msg.sender == owner(), "Only owner or vote manager");
        _;
    }

    modifier onlyDepositManager() {
        require(msg.sender == depositManager, "Only deposit manager");
        _;
    }

    modifier onlyApprovedGauge(address gauge) {
        require(perGaugeApproval[msg.sender] == gauge || msg.sender == depositManager, "Not approved for gauge");
        _;
    }

    /**
        @notice Grant or revoke permission for `caller` to call one or more
                functions on `target` via this contract.
     */
    function setExecutePermissions(
        address caller,
        address target,
        bytes4[] memory selectors,
        bool permitted
    ) external onlyOwner returns (bool) {
        mapping(bytes4 => bool) storage _executePermission = executePermissions[caller][target];
        for (uint256 i = 0; i < selectors.length; i++) {
            _executePermission[selectors[i]] = permitted;
        }
        return true;
    }

    /**
        @notice Set the fee percent taken on all CRV earned through this contract
        @dev CRV earned as fees is periodically added to the contract's locked position
     */
    function setCrvFeePct(uint64 _feePct) external onlyOwner returns (bool) {
        require(_feePct <= 10000, "Invalid setting");
        crvFeePct = _feePct;
        emit CrvFeePctSet(_feePct);
        return true;
    }

    function setVoteManager(address _voteManager) external onlyOwner returns (bool) {
        voteManager = _voteManager;

        return true;
    }

    function setDepositManager(address _depositManager) external onlyOwner returns (bool) {
        depositManager = _depositManager;

        return true;
    }

    function setPerGaugeApproval(address caller, address gauge) external onlyDepositManager returns (bool) {
        perGaugeApproval[caller] = gauge;

        return true;
    }

    /**
        @notice Claim pending 3CRV fees earned from the veCRV balance
                and transfer the fees onward to the fee receiver
        @dev This method is intentionally left unguarded
     */
    function claimFees() external returns (uint256) {
        feeDistributor.claim();
        uint256 amount = feeToken.balanceOf(address(this));

        feeToken.transfer(PRISMA_CORE.feeReceiver(), amount);

        return amount;
    }

    /**
        @notice Lock any CRV balance within the contract, and extend
                the unlock time to the maximum possible
        @dev This method is intentionally left unguarded
     */
    function lockCRV() external returns (bool) {
        uint256 maxUnlock = ((block.timestamp / WEEK) * WEEK) + MAX_LOCK_DURATION;
        uint256 amount = CRV.balanceOf(address(this));

        _updateLock(amount, unlockTime, maxUnlock);

        return true;
    }

    /**
        @notice Mint CRV rewards earned for a specific gauge
        @dev Once per week, also locks any CRV balance within the contract and extends the lock duration
        @param gauge Address of the gauge to mint CRV for
        @param receiver Address to send the minted CRV to
        @return uint256 Amount of CRV send to the receiver (after the fee)
     */
    function mintCRV(address gauge, address receiver) external onlyApprovedGauge(gauge) returns (uint256) {
        uint256 initial = CRV.balanceOf(address(this));
        minter.mint(gauge);
        uint256 amount = CRV.balanceOf(address(this)) - initial;

        // apply fee prior to transfer
        uint256 fee = (amount * crvFeePct) / 10000;
        amount -= fee;

        CRV.transfer(receiver, amount);

        // lock and extend if needed
        uint256 unlock = unlockTime;
        uint256 maxUnlock = ((block.timestamp / WEEK) * WEEK) + MAX_LOCK_DURATION;
        if (unlock < maxUnlock) {
            _updateLock(initial + fee, unlock, maxUnlock);
        }

        return amount;
    }

    /**
        @notice Submit one or more gauge weight votes
     */
    function voteForGaugeWeights(GaugeWeightVote[] calldata votes) external ownerOrVoteManager returns (bool) {
        for (uint256 i = 0; i < votes.length; i++) {
            gaugeController.vote_for_gauge_weights(votes[i].gauge, votes[i].weight);
        }

        return true;
    }

    /**
        @notice Submit a vote within the Curve DAO
     */
    function voteInCurveDao(IAragon aragon, uint256 id, bool support) external ownerOrVoteManager returns (bool) {
        aragon.vote(id, support, false);

        return true;
    }

    /**
        @notice Approve a 3rd-party caller to deposit into a specific gauge
        @dev Only required for some older Curve gauges
     */
    function approveGaugeDeposit(address gauge, address depositor) external onlyApprovedGauge(gauge) returns (bool) {
        ILiquidityGauge(gauge).set_approve_deposit(depositor, true);

        return true;
    }

    /**
        @notice Set the default receiver for extra rewards on a specific gauge
        @dev Only works on some gauge versions
     */
    function setGaugeRewardsReceiver(address gauge, address receiver) external onlyApprovedGauge(gauge) returns (bool) {
        ILiquidityGauge(gauge).set_rewards_receiver(receiver);

        return true;
    }

    /**
        @notice Withdraw LP tokens from a gauge
        @param gauge Address of the gauge to withdraw from
        @param lpToken Address of the LP token we are withdrawing from the gauge.
                       The contract trusts the caller to supply the correct address.
        @param amount Amount of LP tokens to withdraw
        @param receiver Address to send the LP token to
     */
    function withdrawFromGauge(
        address gauge,
        IERC20 lpToken,
        uint256 amount,
        address receiver
    ) external onlyApprovedGauge(gauge) returns (bool) {
        ILiquidityGauge(gauge).withdraw(amount);
        lpToken.transfer(receiver, amount);

        return true;
    }

    /**
        @notice Transfer arbitrary token balances out of this contract
        @dev Necessary for handling extra rewards on older gauge types
     */
    function transferTokens(
        address receiver,
        TokenBalance[] calldata balances
    ) external onlyDepositManager returns (bool) {
        for (uint256 i = 0; i < balances.length; i++) {
            balances[i].token.safeTransfer(receiver, balances[i].amount);
        }

        return true;
    }

    /**
        @notice Execute an arbitrary function call using this contract
        @dev Callable via the owner, or if explicit permission is given
             to the caller for this target and function selector
     */
    function execute(address target, bytes calldata data) external returns (bytes memory) {
        if (msg.sender != owner()) {
            bytes4 selector = bytes4(data[:4]);
            require(executePermissions[msg.sender][target][selector], "Not permitted");
        }
        return target.functionCall(data);
    }

    function _updateLock(uint256 amount, uint256 unlock, uint256 maxUnlock) internal {
        if (amount > 0) {
            if (unlock == 0) {
                votingEscrow.create_lock(amount, maxUnlock);
                unlockTime = uint64(maxUnlock);
                return;
            }
            votingEscrow.increase_amount(amount);
        }
        if (unlock < maxUnlock) {
            votingEscrow.increase_unlock_time(maxUnlock);
            unlockTime = uint64(maxUnlock);
        }
    }
}

// File: Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "IERC20.sol";
import "draft-IERC20Permit.sol";
import "Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: IGaugeController.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

interface IGaugeController {
    function vote_for_gauge_weights(address gauge, uint weight) external;

    function gauge_types(address gauge) external view returns (int128);
}

// File: ILiquidityGauge.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

interface ILiquidityGauge {
    function deposit(uint256 amount, address receiver) external;

    function withdraw(uint value) external;

    function lp_token() external view returns (address);

    function set_approve_deposit(address depositor, bool can_deposit) external;

    function set_rewards_receiver(address receiver) external;
}

// File: PrismaOwnable.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

import "IPrismaCore.sol";

/**
    @title Prisma Ownable
    @notice Contracts inheriting `PrismaOwnable` have the same owner as `PrismaCore`.
            The ownership cannot be independently modified or renounced.
 */
contract PrismaOwnable {
    IPrismaCore public immutable PRISMA_CORE;

    constructor(address _prismaCore) {
        PRISMA_CORE = IPrismaCore(_prismaCore);
    }

    modifier onlyOwner() {
        require(msg.sender == PRISMA_CORE.owner(), "Only owner");
        _;
    }

    function owner() public view returns (address) {
        return PRISMA_CORE.owner();
    }

    function guardian() public view returns (address) {
        return PRISMA_CORE.guardian();
    }
}

// File: IPrismaCore.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.19;

interface IPrismaCore {
    function owner() external view returns (address);

    function guardian() external view returns (address);

    function feeReceiver() external view returns (address);

    function priceFeed() external view returns (address);

    function paused() external view returns (bool);

    function startTime() external view returns (uint256);

    function acceptTransferOwnership() external;

    function setGuardian(address guardian) external;
}